


Arden Syntax overview

Educational material, part 2

Medexter Healthcare
Borschkegasse 7/5
A-1090 Vienna

www.medexter.com

www.meduniwien.ac.at/kpa (academic)



```
logic:  
  let hmi be weight / (size ** 2); // BMI  
  age := currenttime - birth; // AGE  
  if the age is less than 19 years then  
    classification := null;  
  elseif the hmi is less than 18.5 then  
    classification := localized 'under';  
  elseif the hmi is less than 25 then  
    classification := null; // BMI normal range  
  else  
    let the classification be localized 'over';  
  endif;
```

Better care, patient safety, and quality assurance by Medexter, Vienna, Austria

Medical logic modules

Arden Syntax – Structure

- In Arden Syntax, medical knowledge is hierarchically arranged within medical logic modules (MLMs)
- Each MLM represents sufficient knowledge to make at least one single medical decision
- An MLM is stored in a file that has the file extension “.mlm”
- Each MLM is well organized and structured into **categories** and **slots**.
- Categories must appear in a predefined order.
- Each category contains a category-specific set of **slots**, also in a predefined order.

```

maintenance
library
knowledge
resources
  
```

```

maintenance:
  title:                ;;
  mllname:              [required];;
  arden:                [required];;
  version:              [required];;
  institution:          [required];;
  author:               ;;
  specialist:           ;;
  date:                 [required];;
  validation:           [required];;

library:
  purpose:              ;;
  explanation:          ;;
  keywords:             ;;
  citations:            ;;
  links:               ;;

knowledge:
  type:                 data_driven;;
  data:                 ;;;
  priority:             ;;;
  evoke:                ;;;
  logic:                ;;;
  action:               ;;;
  urgency:              ;;;

resources:              [optional]
  default:              ;;;
  language:             ;;;

end:
  
```

Arden Syntax MLM – Maintenance Category

- Purpose
 - To provide general information on the MLM, unrelated to the MLM's medical knowledge.
 - Used for MLM knowledge base maintenance and version control.
- Slots (**bold** means required)
 - **title**: Title of the MLM
 - **mlmname**: Name as referred to by the host system; must have the extension .mlm.
 - **arden**: Arden Syntax version
 - **version**: MLM version
 - **institution**: Institution which created / adopted the MLM.
 - **author**: Author / implementer of the MLM
 - **specialist**: Medical specialist supervising the MLM's medical content.
 - **date**: Date of creation / last update.
 - **validation**: The MLM's current status: *production, research, testing, or expired.*
- **MLMname, institution, and MLM version** are used to identify the MLM.



Arden Syntax MLM – Maintenance Category – Example

```
1 maintenance:
2   title:      SIRS alert notification;;
3   mlmname:    SIRSNotification;;
4   arden:      version 2.9;;
5   version:    1.0;;
6   institution: Medexter Healthcare, Vienna, Austria;;
7   author:     Knowledge engineering group;;
8   specialist: Clinical specialists group;;
9   date:       2016-08-19;;
10  validation: production;;
```

Arden Syntax MLM – Library Category

- Purpose
 - Meant to provide medical background information about the MLM.
 - Contains slots pertinent to maintenance activities related to the MLM's medical knowledge.
- Slots
 - **purpose:** This slot describes the clinical aim of the MLM.
 - **explanation:** Description on how the MLM works.
 - **keywords:** Keywords are descriptive words used for searching through modules. Keywords are delimited by semicolons.
 - **citations:** Citations allow for the documentation of references to relevant used literature.
 - **links:** This slot allows for providing links to sources of information relevant for this MLM (e.g., electronic textbook, teaching cases, web resources).



Arden Syntax MLM – Library Category – Example

```
11 library:
12     purpose:      To generate an alert once SIRS indications are sufficiently
13                   present.
14                   ;;
15     explanation:  The MLM captures input data, then interprets this data
16                   using implemented SIRS rules. For each matched rule, a
17                   positive rule counter is increased by one. If, at the end,
18                   two or more rules matched, an alert is generated.
19                   ;;
20     keywords:     Systemic Inflammatory Response Syndrome
21                   ;;
22     citations:     Definitions For Sepsis And Organ Failure And Guidelines For
23                   The Use Of Innovative Therapies In Sepsis. The Accp/Sccm
24                   Consensus Conference Committee. American College Of Chest
25                   Physicians/Society Of Critical Care Medicine.
26                   Bone R, Balk R, Cerra F, et al.
27                   Chest. 1992;101(6):1644-1655.;;
28     links:        ;;
```

Arden Syntax MLM – Knowledge Category – Example

SIRS Notification

ALERT if ≥ 2 Criteria

Temperature $> 38^{\circ}\text{C}$ (100.4°F) or $< 36^{\circ}\text{C}$ (96.8°F)

and/or

Heart rate > 90 beats per minute

and/or

Respiratory rate > 20 breaths per minute or arterial carbon dioxide tension (PaCO₂) < 32 mm Hg

and/or

White blood cell count ($>12,000/\mu\text{L}$ or $< 4,000/\mu\text{L}$ or $>10\%$ immature [band] forms)

SIRS: systemic inflammatory response syndrome, unsharpness of boundaries not considered

```
17 knowledge:
18     type: data_driven;;
19     data:
20         (temperature,heartRate,respRate,
21          PaCO2,WBcellCount,immatureBand):= Argument;
22     ;;
23     priority: ;;
24     evoke: ;;
25     logic:
26         //Start - Checking SIRS criteria
27         counter:=0;
28
29         if temperature is greater than 38 or temperature is less than 36 then
30             counter:= counter + 1;
31         endif;
32
33         if heartRate is greater than 90 then
34             counter:= counter + 1;
35         endif;
36
37         if respRate is greater than 20 or PaCO2 is less than 32 then
38             counter:= counter + 1;
39         endif;
40
41         if WBcellCount is greater than 12000 or WBcellCount is less than 4000
42         or immatureBand is greater than 10 then
43             counter:= counter + 1;
44         endif;
45
46         if counter is greater than or equal 2 then
47             notification:= localized 'SIRS';
48             conclude true;
49         endif;
50         //End - Checking SIRS criteria
51
52         conclude true;
53     ;;
54     action:
55         return notification;
56     ;;
57     urgency: ;;
```

Use Case: Hypoglycemia at intensive care units

- Hypoglycemia may seriously harm.
- If patient is unconscious, it is difficult to notice.
- The PDMS (patient data management system) should actively notify the physician:

If glucose is less than 50mg/dl, then send an SMS message to the physician.



```
DATA:  
LET glucose BE READ {...glucose...};  
LET physician_DECT BE DESTINATION {sms:26789};  
  
LOGIC:  
IF LATEST glucose IS LESS THAN 50 THEN  
    CONCLUDE true;  
ENDIF;  
  
ACTION:  
WRITE „Warning...” AT physician_DECT;
```



CONCLUDE TRUE
→ Do something

Arden Syntax MLM – Resources Category

- Purpose
 - Allows for defining localized, different language messages.
 - Specify the textual resources from which the localized messages are taken.
 - The Resources Category is optional.
- Slots (**bold** means required)
 - **default**: Defines the default language to be used.
 - **language**: One language slot for each language to be used. Language codes are defined either as 2-character ISO 639-1 codes or as combination of this language code and a 2-character ISO 3166-1 geographical code.



Arden Syntax MLM – Resources Category – Example

```
59 resources:  
60     default: en;;  
61     language: en  
62         'SIRS': "Alert for SIRS";  
63     ;;  
64     language: de  
65         'SIRS': "Alert für SIRS";  
66     ;;
```

Some operators


Arden Syntax operators—I

Arden Syntax	
Type	Examples
Assignment	:=
List Operators	“, Merge, Sort, Add ... To ... [At ...], Remove ... From ...
Where Operator	where
Logical Operators	And, Or, Not
Simple Comparison Operators	=, <>, <, <=, >, >=
Is Comparison Operators	Is [not] Equal, Is [not] Less Than, Is [not] Greater Than, Is [not] Less Than or Equal, Is [not] Greater Than or Equal, Is [not] Within ... To, Is [not] Within ... Preceding, Is [not] Within ... Following, Is [not] Within ... Surrounding, Is [not] Within ...Past, Is [not] Within ...Same Day As, Is [not] Before, Is [not] After, Is [not] In, Is [not] Present, Is [not]Null, Is [not] Boolean, Is [not] Number, Is [not] String, Is [not] Time, Is [not] Time of day, Is [not]Duration, Is [not] List, [not] In, Is [not] , Is [not] Object, Is [not] <Object-Type>, Is [not] Fuzzy, Is [not] Crisp
Occur Comparison Operators	Occur [not] Equal, Occur [not] Within ... To, Occur [not] Within ... Preceding, Occur [not] Within ... Following, Occur [not] Within ... Surrounding, Occur [not] Within Past, Occur [not] Within Same Day As, Occur [not] Before Occur [not] After, Occur [not] At
String Operators	, Formatted with, String ..., Matches Pattern, Length, Uppercase, Lowercase, Trim [Left Right], Find ... [in] String ... [starting at ...], Substring ... Characters ... [starting at ...] from ..., Localized
Arithmetic Operators	+, -, *, /, **
Temporal Operators	After, Before, Ago, From, Time of day [of], Day of week [of], Extract Year, Extract Month, Extract Day, Extract Hour, Extract minute, Extract second, Replace Year [of] ... With, Replace Month [of] ... With, Replace Day [of] ... With, Replace Hour [of] ... With, Replace Minute [of] ... With, Replace Second [of] ... With
Duration Operators	Year, Month, Week, Day, Hour, Minute, Second

Arden Syntax operators—II

Arden Syntax	
Type	Examples
Aggregation Operators	Count, Exist, Average, Median, Sum, Stddev, Variance, Minimum, Maximum, Last, First, Any [IsTrue], All [AreTrue], No [IsTrue], Latest, Earliest, Element, Extract Characters ..., Seqto, Reverse, Index Extraction
Query Aggregation Operators	Nearest ... From, Index Nearest ... From, Index Of ... From ..., At Least ... [IsTrue AreTrue] From ..., At Most ... [IsTrue AreTrue] From ..., Slope
Transformation Operators	Minimum ... From, Maximum ... From, First ... From, Last ... From, Sublist ...Elements [Starting at ...] From, Increase, Decrease, % Increase, % Decrease, Earliest ... From, Latest ... From, Index Extraction Transformation Operators
Query Transformation Operator	Interval
Numeric Function Operators	Arccos, Arcsin, Arctan, Cosine, Sine, Tangent, Exp, Log, Log10, Int, Floor, Ceiling, Truncate, Round, Abs, Sqrt
Time Function Operator	Time, Time of Objects, Attime
Object Operators	Dot, Clone, Extract Attribute Names ..., Attribute ... From ...
Fuzzy Operators	Fuzzy Set, Fuzzyfied By, Defuzzified, Application [of]..., Applicability of Objects, ...
Type Conversion Operator	As Number, As Time, As String, As Truth Value

The ArdenSuite



AVAILABLE FOR:



Windows



Linux



Mac OS



Docker Container



Amazon Cloud



Microsoft Azure